

## Optimize My Stock Portfolio!

### A Case Study with Three Different Estimates of Risk

Aric LaBarr, Institute for Advanced Analytics at North Carolina State University

#### ABSTRACT

People typically invest in more than one stock to help diversify their risk. These stock portfolios are a collection of assets that each have their own inherent risk. If you know the future risk of each of the assets, you can optimize how much of each asset to keep in the portfolio. The real challenge is trying to evaluate the potential future risk of these assets. Different techniques provide different forecasts, which can drastically change the optimal allocation of assets. This talk presents a case study of portfolio optimization in three different scenarios - historical standard deviation estimation, capital asset pricing model (CAPM), and GARCH-based volatility modeling. The structure and results of these three approaches will be discussed.

#### INTRODUCTION

A portfolio is a collection of assets in which investors decide how much money they will invest in each asset to try and maximize the portfolio's performance over a certain period of time. For example, imagine you are investing in a portfolio containing the following four technology company stocks – Apple, Google, Microsoft, and Ebay. How much money should you invest in each of these stocks to build your portfolio? This decision is not a trivial one because each asset in a portfolio contributes to the overall performance of the portfolio in two ways – return and risk. Return is the percentage growth in the value of an asset and investors want to maximize it. However, in maximizing return, you also want to minimize the risk or the variability associated with the returns on the asset. This variability is often called volatility. The goal is to maximize the return while minimizing the risk. Higher returns are not typically achieved without higher risk to obtain those returns.

Continuing on the previous example, the return and risk for a four stock portfolio is the following:

$$\text{Return} = p_1 r_1 + p_2 r_2 + p_3 r_3 + p_4 r_4$$
$$\text{Risk} = \sum_j \sum_k p_j \times \sigma_{j,k} \times p_k = p_1 \sigma_{1,1} p_1 + p_1 \sigma_{1,2} p_2 + \cdots + p_4 \sigma_{4,4} p_4$$

where  $r_i$  is the return for stock  $i$ ,  $p_i$  is the proportion of wealth invested in stock  $i$ , and  $\sigma_{j,k}$  is the covariance between stock  $j$  and stock  $k$ . For simplification, assume the proportions of wealth must sum to one (i.e.  $p_1 + p_2 + p_3 + p_4 = 1$ ). This paper does not cover the more complicated scenario relaxing this assumption. In any portfolio optimization problem you have two choices – maximize return for a given risk, or the more typical, minimize risk for a given return. The second choice is a straight-forward optimization problem where risk is the objective function you are trying to minimize and return is a constraint. The OPTMODEL procedure in SAS® is designed to handle these cases:

```
proc optmodel;

/* Declare Sets and Parameters */
set <str> Assets1, Assets2, Assets3;
num Covariance{Assets1,Assets2};
num Mean{Assets1};

/* Read in SAS Data Sets */
read data Cov into Assets1=[_NAME_];
read data Cov into Assets2=[_NAME_] {i in Assets1}
    <Covariance[I,_NAME_]=col(i)>;
```

```

read data Mean into Assets3=[_NAME_] {i in Assets1} <Mean[i]=col(i)>;

/* Declare Variables */
var Proportion{Assets1} >= 0 init 0;

/* Declare Objective Function */
min Risk = sum{i in Assets1}
    (sum{j in Assets1}Proportion[i]*Covariance[i,j]*Proportion[j]);

/* Declare Constraints */
con Return: 0.0005 <= sum{i in Assets1}Proportion[i]*Mean[i];
con Sum: 1 = sum{i in Assets1}Proportion[i];

/* Call the Solver */
solve;

/* Print Solutions */
print Covariance Mean;
print Proportion;

quit;

```

Looking back at historical data, this is where the analysis would end. However, investors seek to optimize the allocation of assets for future returns and future values of risk. Therefore, we must estimate future values of return and risk for inputs to the optimization. We assume the future value of returns for each stock are the historical average returns for each stock. This paper discusses three different techniques to estimate the future values of risk – historical variance (or standard deviation), the capital asset pricing model (CAPM), and generalized autoregressive conditional heteroscedasticity (GARCH) volatility estimation. From these calculations, this paper compares the optimal allocation of assets under each of these techniques.

### 3 DIFFERENT ESTIMATES OF RISK

Estimation of volatility for assets is not easy. A variety of techniques are used to estimate volatility for different assets. Three of the most popular ways of estimating volatility are using historical variances (or standard deviations), the CAPM, and GARCH modeling.


#### HISTORICAL VARIANCE APPROACH

It is reasonable to assume that volatility in historical data helps estimate future values of volatility. However, it is not reasonable to assume that volatility stays constant across time. Understanding the characteristics of financial assets helps develop estimation methods for volatilities. Figure 1 displays the daily returns of Microsoft stock over time.

From Figure 1 we can see that volatility varies over time instead of remaining constant. The volatilities also exhibit some persistence, or dependence over time. This appears as a clustering effect where large (or small) values of returns are followed by large (or small) returns of either sign. To account for the changing volatility over time, calculate the historical variance of the asset using a rolling window calculation. In this calculation, estimate the volatility by the variance of the  $m$  most recent returns:

$$\sigma_t^2 = \frac{1}{m-1} \sum_{i=1}^m (r_{t-i} - \bar{r})^2$$

The value of  $m$  should be large enough to capture the important recent returns, but small enough to not let returns in the distant past negatively influence the calculation. Table 1 shows different historical variance estimates using different windows of time. The values are small due to the small variation in day



to day returns, but the comparison between them is the important piece. Sometimes investors look at the standard deviations instead of the variances to put the calculation in the same scale as the original data.

WINDOW OF TIME	ESTIMATE OF MSFT VOLATILITY	ESTIMATE OF MSFT / AAPL COVARIANCE
1/1/2008 – 12/31/2016	0.000329	0.000175
1/1/2011 – 12/31/2016	0.000216	0.000089
1/1/2013 – 12/31/2016	0.000227	0.000081
1/1/2016 – 12/31/2016	0.000205	0.000104

**Table 1. Historical Variance/Covariance Estimates of MSFT Volatility**

Picking these windows of time involves investigating the daily returns to try to see where the data seems stable, which is not always easy. For simplicity, this paper assumes a window of one year for each historical estimate of volatility of the stocks in the portfolio.

The variance is the volatility of the individual stock, but risk also depends on the relationship, or covariance, between stocks. Similar to the variance calculation, covariance calculations are over a rolling window of time. The covariance between the returns on stock  $j$  and stock  $k$  is the following:

**Figure 1. Daily Returns for Microsoft Stock from 1/1/2008 - 12/31/2016**

$$\sigma_{j,k} = \frac{1}{m-1} \sum_{i=1}^m (r_{j,t-i} - \bar{r}_j)(r_{k,t-i} - \bar{r}_k)$$

The value of  $m$  should be large enough to capture the important recent returns, but small enough to not let returns in the distant past negatively influence the calculation. Table 1 shows different historical covariance estimates using different windows of time. The values are small due to the small variation in day to day returns, but the comparison between them is the important piece. Similar to variance, this

paper assumes a window of one year for each historical estimation of covariance of the stocks in the portfolio.

The CORR procedure in SAS easily calculates the variances and covariances for a set of variables:

```
/* Calculate the Covariances Between Stocks for 2016 */
proc corr data=Stocks (where=(Date>='01jan2016'd)) cov out=Corr;
  var MSFT_r AAPL_r GOOGL_r EBAY_r;
run;

/* Save Covariance Matrix to Own Data Set */
data Cov;
  set Corr;
  where _TYPE_ = 'COV';
run;

/* Save Average Returns to Own Data Set */
data Mean;
  set Corr;
  where _TYPE_ = 'MEAN';
run;
```

The above code along with the previous used code from PROC OPTMODEL optimizes a stock portfolio using the historical variances approach.

## CAPITAL ASSET PRICING MODEL APPROACH

Another approach to estimating the risk of a portfolio is the Capital Asset Pricing Model (CAPM) approach. The CAPM equation relates the expected excess return of an asset to the expected excess return of the market in which the asset belongs:

$$E(r_i) - r_f = \delta_i (E(r_M) - r_f)$$

where  $r_i$  is the return on the asset of interest,  $r_f$  is the risk-free rate of return (ex. 3-month T-bill rate), and  $r_M$  is the return of the market portfolio (ex. return of market index typically used). The relationship between the individual asset and the market,  $\delta_i$ , informs us of the risk of the asset. Assuming that higher returns involve more risk, as mentioned earlier, if the value of  $\delta_i > 1$  then the asset is perceived riskier than the market. If the value of  $\delta_i < 1$  then the market is perceived as riskier. If the market and asset have the same risk, then  $\delta_i = 1$ .

Asset managers use single and multiple factor models built on the CAPM. A factor model describes the relationship between the expected return of an asset (or portfolio of assets) and a set of market risk factors. A single factor model only relates a single risk factor using simple linear regression, while a multiple factor model relates multiple risk factors in multiple linear regression. For simplicity, this paper will use a single factor model based on the CAPM – a simple linear regression relating an individual asset to the market portfolio:

$$r_{i,t} = \alpha_i + \beta_i X_t + \varepsilon_{i,t}$$

$$\varepsilon_{i,t} \sim iid(0, \sigma_i^2)$$

where  $r_{i,t}$  is the return of the asset of interest during period  $t$ ,  $\alpha_i$  is the excess return relative to the index,  $X_t$  is the return on the market index during period  $t$ , and  $\varepsilon_{i,t}$  is the error in the factor model. The variance of the errors,  $\sigma_i^2$ , in the model is an estimate of the specific risk of the asset itself, while  $\beta_i^2 \sigma_X^2$  is an estimate of the market risk, also called the systematic risk. We perform this calculation for every asset in our portfolio to estimate the risk of the portfolio.

Essentially the factor model breaks down our estimate of risk into two pieces – the risk of the asset and the risk of the market. We cannot control the risk of the market as that is inherent in any investment portfolio. However, we control the risk of the portfolio by determining how much of each asset we want based on the individual estimates of the asset's risk. For our example with four stocks, calculate the single factor model above for each of the 4 stocks in the portfolio with  $X_t$  representing the Dow Jones Industrial Average (DJIA) index representing the market portfolio. Asset managers prefer to use three to five years of data for these calculations, so we use 3 years in this paper. The return and risk of the four stock portfolio are defined as follows:

$$Return = \sum_{i=1}^4 p_i \hat{\alpha}_i + \bar{X} \left( \sum_{i=1}^4 p_i \hat{\beta}_i \right)$$

$$Risk = \sum_{i=1}^4 p_i \sigma_{\varepsilon_i}^2 + \sigma_X^2 \left( \sum_{i=1}^4 p_i \hat{\beta}_i \right)^2$$

where  $p_i$  is the proportion of wealth invested in stock  $i$ ,  $\hat{\alpha}_i$  and  $\hat{\beta}_i$  are estimated from each factor equation,  $\sigma_X^2$  is the variance of the market index, and  $\sigma_{\varepsilon_i}^2$  is the variance of the errors from each factor model.

The REG and MEAN procedures in SAS easily calculates the needed information and regressions for a set of variables:

```
/* Calculate 2014-2016 DJIA Mean and Variance for CAPM */
proc means data=Stocks(where=(Date>='01jan2014'd)) mean var;
    var DJI_r;
    output out=DJIX var=Var mean=Mean;
run;

/* Calculate CAPM Regressions for Stocks for 2014-2016 */
proc reg data=Stocks(where=(Date>='01jan2014'd)) outest=Coef;
    MSFT: model MSFT_r = DJI_r;
    AAPL: model AAPL_r = DJI_r;
    GOOGL: model GOOGL_r = DJI_r;
    EBAY: model EBAY_r = DJI_r;
run;
quit;
```

The above code produces SAS data sets containing all the needed information for the optimization. The previous PROC OPTMODEL code cannot be used as the objective function and constraint have changed. The following is the updated PROC OPTMODEL code handling the adjustments to the optimization:

```
/* Optimization of Assets - CAPM */
proc optmodel;

    /* Declare Sets and Parameters */
    set <str> Assets1;
    num Alpha{Assets1};
    num Beta{Assets1};
    num Sigma{Assets1};
    num MeanX;
    num VarX;

    /* Read in SAS Data Sets */
    read data Coef into Assets1=[_DEPVAR_] Alpha=col("Intercept");
```

```

read data Coef into Assets1=[_DEPVAR_] Beta=col("DJI_r");
read data Coef into Assets1=[_DEPVAR_] Sigma=col("_RMSE_");
read data DJIX into MeanX=col("Mean");
read data DJIX into VarX=col("Var");

/* Declare Variables */
var Proportion{Assets1} >= 0 init 0;

/* Declare Objective Function */
min Risk = sum{i in Assets1}Proportion[i]*Sigma[i]**2
          + VarX*(sum{i in Assets1}Proportion[i]*Beta[i])**2;

/* Declare Constraints */
con Return: 0.0005 <= sum{i in Assets1}Proportion[i]*Alpha[i]
              + MeanX*(sum{i in Assets1}Proportion[i]*Beta[i]);
con Sum: 1 = sum{i in Assets1}Proportion[i];

/* Call the Solver */
solve;

/* Print Solutions */
print Alpha Beta Sigma MeanX VarX;
print Proportion;

quit;

```

There are two important assumptions for this approach estimating the amount of each asset in the portfolio. The first assumption is that the linear regression model appropriately estimates the relationship between the market index and the portfolio. The second, and harder assumption to make, is independence between the specific risks of each asset. This second assumption implies that all assets are subject to the market risk, but their individual risks do not influence each other. This difficult assumption leads to the last method of calculating risk – the GARCH model approach.

## GARCH VOLATILITY ESTIMATION APPROACH

Generalized autoregressive conditional heteroscedasticity (GARCH) models were created by Engle (1982) with the ARCH model and generalized by Bollerslev (1986) into the GARCH model to estimate volatility. Instead of a weighting scheme, Engle uses an autoregressive time series approach to account for persistence in volatility estimation. He also assumes that the volatilities are conditional on each other over time. Engle's model is defined as the following:

$$\sigma_t^2 = \alpha_0 + \alpha_1 r_{t-1}^2$$

In this model, the intercept  $\alpha_0$  is positive, and the parameter coefficient  $\alpha_1$  is nonnegative so that the volatility estimation is positive. The returns here in Engle's model are also assumed to be normally distributed with a mean of zero and variance of  $\sigma_t^2$ , conditional on all information up to the current time. Sometimes this model is referred to as the normal ARCH model for this reason. Of course this model can easily extend to include many lags of returns since it would be unreasonable to assume that today's volatility only depends on yesterday's returns. This is called the ARCH(q) model:

$$\sigma_t^2 = \alpha_0 + \alpha_1 r_{t-1}^2 + \cdots + \alpha_q r_{t-q}^2 = \alpha_0 + \sum_{i=1}^q \alpha_i r_{t-i}^2$$

However, in practice, we often need large values of  $q$  to accurately model real world data. Bollerslev avoids the problem of large values of  $q$  in Engle's ARCH model by generalizing the ARCH(q) model in much the same way as an autoregressive (AR) model extends to the autoregressive moving average (ARMA) model. The GARCH model allows for a longer memory process with more flexibility. The GARCH

model stills assumes normality as Engle's model did, but now includes previous estimates of volatility in the calculation:

$$\sigma_t^2 = \alpha_0 + \alpha_1 r_{t-1}^2 + \beta_1 \sigma_{t-1}^2$$

Similar to the ARCH model that be extended to the ARCH(q) model, the GARCH model can be extended to the GARCH(q, p) model as follows:

$$\sigma_t^2 = \alpha_0 + \sum_{i=1}^q \alpha_i r_{t-i}^2 + \sum_{i=1}^p \beta_i \sigma_{t-i}^2$$

Just like with Engle's model, the intercept must be positive, while the coefficients must be nonnegative to ensure that the volatility estimate is positive. In many cases,  $q = p = 1$ , is an adequate model to fit real world data.

This approach doesn't use historical estimates of variance for each stock or historical relationships with the market index, but instead forecasts future values of volatility to use for optimizing a portfolio of stocks. Risk managers typically use 1-2 years of stock data for GARCH model calculations, so this paper uses 2 years. The AUTOREG procedure in SAS estimates a variety of GARCH models. The following PROC AUTOREG code estimates the basic GARCH(1,1) model described above:

```
/* Estimate GARCH(1,1) Model - MSFT */
proc autoreg data=Stocks(where=(Date>='01jan2015'd));
  model msft_r = / garch=(p=1, q=1) method=ml;
  output out=garch_n ht=predicted_var;
run;
```

Countless adaptations have been made to the GARCH framework of models over the past three decades. Two important adaptations are the t-GARCH model by Bollerslev (1987) and the QGARCH model by Engle and Ng (1993). Bollerslev (1987) proposes one of the first extensions to the ARCH/GARCH framework by noting that financial returns are typically not normally distributed and are conditionally leptokurtic (fatter tails than a normal distribution). He proposes an underlying t-distribution instead of a normal distribution for the returns. SAS calculates the t-GARCH model using the following distribution option in the model statement of PROC AUTOREG:

```
/* Estimate t-GARCH(1,1) Model - MSFT */
proc autoreg data=Stocks(where=(Date>='01jan2015'd));
  model msft_r = / garch=(p=1, q=1) dist=t method=ml;
  output out=garch_t ht=predicted_var;
run;
```

However, both the normal and t-distribution based GARCH models assume symmetry to financial returns. Not all instances have this quality. Nelson (1991) and later Engle and Ng (1993) propose models to account for the leverage effect in financial returns. The leverage effect is when volatility increases more when a return is negative compared to when a return is positive. This implies people are risk averse and will tend to react differently to negative returns than positive ones. Engle and Ng created the quadratic GARCH (QGARCH) model accounting for this:

$$\sigma_t^2 = \alpha_0 + \sum_{i=1}^q \alpha_i (r_{t-i} - \psi_i)^2 + \sum_{i=1}^p \beta_i \sigma_{t-i}^2$$

where  $\psi_i$  is a correction shifting the returns centers away from zero to account for the leverage effect. SAS calculates the QGARCH model using the following type option in the model statement of PROC AUTOREG:

```

/* Estimate QGARCH(1,1) Model - MSFT */
proc autoreg data=Stocks(where=(Date>='01jan2015'd));
  model msft_r = / garch=(p=1, q=1, type=QGARCH) method=ml;
  output out=qgarch ht=predicted_var;
run;

```

Both models above can be combined in PROC AUTOREG using both the distribution and type options together.

All four of these models – GARCH(1,1), t-GARCH(1,1), QGARCH(1,1), t-QGARCH(1,1) – are fit for each of the four stocks in our example. The model with the best values of AIC, SBC, and Log-Likelihood are chosen for each and displayed in Table 2. The quadratic GARCH model with a t-distribution is the best model according to AIC, SBC, and Log-Likelihood for each of the four stocks in our example.

STOCK	MODEL	AIC	SBC	LOG- LIKELIHOOD
MSFT	Normal GARCH	-2738	-2721	1373
	t-distribution	-2884	-2863	1447
	Quadratic	-2738	-2717	1374
	<b>t-distribution &amp; Quadratic</b>	<b>-3473</b>	<b>-3448</b>	<b>1743</b>
AAPL	Normal GARCH	-2775	-2757	1391
	t-distribution	-2819	-2798	1415
	Quadratic	-2797	-2776	1403
	<b>t-distribution &amp; Quadratic</b>	<b>-3406</b>	<b>-3380</b>	<b>1709</b>
GOOGL	Normal GARCH	-2840	-2824	1424
	t-distribution	-2913	-2892	1462
	Quadratic	-2839	-2818	1425
	<b>t-distribution &amp; Quadratic</b>	<b>-3494</b>	<b>-3468</b>	<b>1753</b>
EBAY	Normal GARCH	-2614	-2597	1311
	t-distribution	-2811	-2794	1410
	Quadratic	-2615	-2594	1313
	<b>t-distribution &amp; Quadratic</b>	<b>-3387</b>	<b>-3361</b>	<b>1699</b>

**Table 2: GARCH Model Comparison and Selection for MSFT, AAPL, GOOGL, and EBAY**

These GARCH models only forecast future values of the variance, not the covariance. In recent years researchers developed multivariate extensions of the GARCH model, however, the complication of these models grows as a simple four asset portfolio now has four variances and six covariances to forecast. For simplicity this paper will use historical calculations of covariance for the forecasted values. This approach assumes that although volatility within a stock fluctuates frequently, the relationship between stocks remains more stable over time. Similar to the historical variance approach, PROC CORR calculates the historical covariance matrix between the four stocks. However, the variances (the diagonal in the matrix) is replaced with the forecasted values of variance from the GARCH models instead of the historical values:



```

/* Calculate the Covariances Between Stocks for 2015-2016 */
proc corr data=SGF_stocks(where=(Date>='01jan2015'd)) cov out=Corr;
    var msft_r aapl_r googl_r ebay_r;
run;

/* Save Covariance Matrix to Own Data Set */
data Cov;
    set Corr;
    where _TYPE_='COV';
run;

/* Save Average Returns to Own Data Set */
data Mean;
    set Corr;
    where _TYPE_='MEAN';
run;


/* Replace Variances in Matrix to Forecasted Values Saved as MACROS */
data Cov;
    set Cov;
    if _NAME_="msft_r" then msft_r=&msft_pred;
    if _NAME_="aapl_r" then aapl_r=&aapl_pred;
    if _NAME_="googl_r" then googl_r=&googl_pred;
    if _NAME_="ebay_r" then ebay_r=&ebay_pred;
run;;

```

The above code along with the original code from PROC OPTMODEL at the beginning of the paper optimizes a stock portfolio using the GARCH modeling approach.

## OPTIMAL ALLOCATION UNDER EACH ESTIMATE OF RISK

Each of the above techniques provides a different approach to estimating future values of risk resulting in drastically different allocations of assets in the four stock portfolio. Assume a simplistic trading strategy



**Figure 2: Optimal Allocation of Stocks Under Each Strategy**

where you optimize the portfolio at the close of business on December 30, 2016 and leave the portfolio allocation alone for 2 business weeks until January 13, 2017.

Figure 2 displays the proportion of wealth that is in each asset of the portfolio under the 2 week trading strategy. The GARCH modeling approach forecasts different volatilities across the 2 week period. The allocation assumes the median volatility across the 2 week forecast window.

The two approaches based more on historical estimation of data – the historical variance and CAPM approaches – lean heavily on MSFT as an investment. The GARCH approach however suggests little investment in MSFT. What does this imply? This implies historically MSFT has been a good investment in terms of volatility and return, however, doesn't forecast to be as good of an investment in the future. The CAPM approach also puts almost all of the asset allocation (~99%) into just two stocks – MSFT and GOOGL. The historical variance and GARCH approaches split the allocation a little more across the four stocks, but only the GARCH approach actually invests in all four in any significant way.

Which strategy was correct? Or more accurately, which strategy provides the best return? Since variances are not observed as the price itself, accuracy measures aren't as valuable. We can observe which strategy provided the greatest value. Table 3 shows the two week returns from the portfolios optimized from each strategy. The sample size is small, but the GARCH approach's stock portfolio provided the best return.

STRATEGY	RETURN
HISTORICAL	1.13%
CAPM	0.88%
GARCH	2.43%

**Table 3: 2 Week Return From Each Portfolio Strategy**

## CONCLUSION

A variety of different ways to decide how to optimize a stock portfolio exist because of the numerous ways of estimating risk. Some estimates of risk rely on past values of risk repeating themselves into the future, while others rely on forecasting future values of risk. Some estimates of risk try to isolate the variability of individual assets from the market variability, while others just role up all the variability together into risk. Three popular ways of evaluating risk are the historical variance approach, the capital asset pricing model (CAPM), and the generalized autoregressive conditional heteroscedasticity (GARCH) model. This paper is not designed to explicitly say one way of estimating risk is always better than another. Instead, this paper tries to help the reader understand that risk can be calculated differently and to approach a problem from a variety of angles. Let the data tell the story, but give the data many different options of the story it can tell.

## REFERENCES

- Engle, Robert. Autoregressive Conditional Heteroscedasticity with Estimates of the Variances of United Kingdom Inflation. *Econometrica*, 987-1007, 1982.
- Bollerslev, Tim. Generalized autoregressive heteroscedasticity. *Journal of Econometrics*, 31:307-327, 1986.
- Bollerslev, Tim & Engle, Robert. Modeling the persistence of conditional variances. *Econometric Reviews*, 5:1-50, 1986.
- Bollerslev, Tim. A conditionally heteroscedastic time series model for speculative prices and rates of return. *The Review of Economics and Statistics*, 69:542-547, 1987.
- Nelson, Daniel. Conditional heteroscedasticity in asset returns: A new approach. *Econometrica*, 59:347-370, 1991.

Engle, R. F. and Ng, V. K. Measuring and Testing the Impact of News on Volatility. *Journal of Finance*, 48, 1749–1778, 1993.

Alexander, Carol. 2008. *Practical Financial Econometrics: Market Risk Analysis II*. Pp. 89-199. West Sussex, England: John Wiley & Sons Ltd.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Aric LaBarr  
Institute for Advanced Analytics at North Carolina State University  
919-513-2957  
aric\_labarr@ncsu.edu  
<http://www4.ncsu.edu/~adlabarr/>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.